

A Case Study in Ada2012
Implementing a Static Http Server, Axios
Prepared for SOEN6411,
and presented to Dr. Constantinides

Simon J. Symeonidis
5887887

November 25, 2013

Contents

1	Abstract	1
1.1	Disclaimer	1
1.2	The name ‘Axios’	1
2	Topics of interest	2
2.1	Concurrency	2
2.2	Sockets	2
2.3	Cross Platform Filesystem Support	2
2.4	Protocol Implementation	2
2.4.1	Overview of Http Protocol	2
2.4.2	HEAD	3
2.4.3	GET	4
2.4.4	POST	4
2.4.5	PUT	4
2.4.6	DELETE	4
2.5	First Steps	4
2.5.1	Deployment View	5
3	Implementation	6
3.1	Defining the required GNAT Project File Specification	6
3.1.1	GNAT Project File	6
3.2	Components	8
3.2.1	Main Entry Point	8
3.2.2	Listener	9
3.2.3	Transaction Handlers	13
3.2.4	Request Helpers	14
3.2.5	Response Helpers	16
3.2.6	File Utils	17
3.3	Testing	19
4	Demonstration	20
5	Appendix	22
5.1	Networking	22
5.2	Task Oriented Concurrency Model	24
5.2.1	Simple Concurrent Program	24
5.3	Text Manipulation	26
5.3.1	Splitting	26
5.3.2	Regular Expressions	26
5.4	Random Notes	26
5.4.1	ADS / ADB file location	26
5.4.2	Sanity Checking	26
5.4.3	Using Telnet to connect to a Http Server	26
5.4.4	Using ncat to dump packets to a binary file	27

List of Figures

1	Deployment Diagram	5
2	Overall organization of software components of Axios	8

3	Using a browser to access the server	20
4	Server output when receiving requests	21

List of Tables

1	HTTP Method Indempodency	3
2	Directory Structure Specification for GPR File	6

Listings

1	Header Definition	3
2	Html Header	3
3	Http Response	3
4	Http Get Request	4
5	GNAT Project File	6
6	Directory Structure	7
7	Main Entry Point	8
8	Simple Listener Specification	9
9	Simple Listener Body	10
10	Transaction Handler Specification	13
11	Transaction Handler Implementation	13
12	Request Helper Specification	14
13	Request Helper Implementation	14
14	Response Helper Specification	16
15	Response Helper Implementation	16
16	File Utils Specification	17
17	Request Helper Implementation	18
18	Simple Listener	22
19	Simple Sender	23
20	VCV Get Http Client	23
21	Task Types in Ada	24
22	Output for the Task Types in Ada	25
23	Telnet to an Http Server	26
24	Using ncat to dump packets	27
25	Using ncat a test script and xxd to hex dump	27
26	One liner to hex dump dynamically	27

Abstract

This is a small case study in the 2012 release of the Ada programming language. The motivation of this case study is to expose one to the broadest topics of the Ada programming language via a detailed example. The topic I have chosen is to write up a very simple web server. A web server typically, has to deal with the following aspects:

- Concurrency
- Sockets, and messaging
- The Protocol (in this case **http**)
- Different Filesystems support

This document is hoped to help the curious reader to learn the aspects of the language with a little more ease, introduce the http protocol slightly, and provide references where possible to direct to sources for more exploration. I also record personal notes.

There is a core aim, to present the aforementioned array of topics to the reader, in a project that is not too big, and intimidate the beginner, and not too small to not be of any learning use. We try and achieve this balance by reducing the lines of code and sacrificing aspects of software architecture¹ in order achieve this.

1.1 Disclaimer

As previously stated, this is a case study that is worked on by a complete beginner in the Ada2012 programming language. You might find that this document is referencing a lot of sources, in a collage fashion. That may be a true fact, but the purpose of this document is not to shed new light to a problem. This document is aimed as a tutorial and reference notes to the programming language, for the beginners and intermediate programmers. Please also consider using the excellent references, as they have provided a lot of help in this case study, and can be used more to broaden ones' knowledge.

1.2 The name 'Axios'

Axios stands for "Ada eXtra ineffectual Object Oriented Server". And it is just that!

¹In the code base, you might notice some aspects that could be factored out, or areas where coupling could be reduced. This will be done on purpose in order to reduce the amount of indirection, and ease the quick navigation, reading and understanding for the reader.

Topics of interest

Section 2

Here we discuss in a little more detail the previously outlined topics. These topics offer us a concise and detailed tests to investigate the the Ada 2012 programming language, along with other tools that can be used to supplement and extend future projects. We will be looking at the following aspects:

- The Ada2012 programming language (Language structure, strengths, weaknesses, etc)
- The Ada2012 Toolchain The **gnat** compiler, the project manager files (gpr)
- AUnit for a unit testing platform

2.1 Concurrency

Ada is considered to have a good concurrency support. We see examples later in the code listings, and some more specific (yet unpractical) examples in the appendix.

2.2 Sockets

Sockets are a good topic in this case study, since sockets are OS dependent therefore will provide some insight on how the language and application will behave on different platforms.

2.3 Cross Platform Filesystem Support

To programmers that have used languages that do not require virtual machines, concerns for cross platform availability of an application, more often arise. This case study will use some simple filesystem operations to indicate what we may and may not do with the language. We will investigate how:

- Location to resources are handled
- Permissions for inodes are handled

2.4 Protocol Implementation

A topic that I have came across in programming languages that is somewhat interesting to see in implementation, are protocols. Notably **Erlang** makes this a breeze. Investigating this in Ada will be interesting.

For this small project, we'll be implementing 5 parts of the Http protocol. Namely the HEAD, GET, POST, PUT and DELETE methods. We refer to the RFC2616 manual [8] for the definition of behaviours of these methods. Once they are implemented in this case study using Ada, the project shall be tested by using any mainstream browser such as *Firefox* or *Chromium*.

For the reader we extract the definitions and behaviour of these four methods.

2.4.1 Overview of Http Protocol

The Http protocol is a text, request-response protocol. The RFC2616 specification is quite detailed, but we will be implementing a very small subset of the specification for this case study.

The Http version we are concerned with is **1.1**. There exists indempotent and non indempotent methods in the Http specification.

GET is described as an indempotent method. That is, if a request is being made with the same specific set of parameters, then the response should be the same each time that request is repeated. For example, if we had a function such as $f(x) = x + 1$, then each time we provided $x = 1$ we should always retrieve '2'. We present the following table, which describes the methods we will be implementing:

Method	Type
HEAD	Indemponent
GET	Indemponent
POST	Non-Indemponent
PUT	Indemponent
DELETE	Indemponent

Table 1: HTTP Method Indempodency

2.4.2 HEAD

This Http method requests the header fields of an html document. The message body is not to be retrieved. Here is an example of what http headers look like. The definition is taken straight out of the manual:

```

1 message-header = field-name ":" [ field-value ]
2 field-name     = token
3 field-value    = *( field-content | LWS )
4 field-content  = <the OCTETs making up the field-value
5                and consisting of either *TEXT or combinations
6                of token, separators, and quoted-string>
```

Listing 1: Header Definition

Here is a more practical example. These headers are retrieved from *www.google.ca*:

```

1 HTTP/1.1 302 Found
2 Location: http://www.google.ca/
3 Cache-Control: private
4 Content-Type: text/html; charset=UTF-8
5 Date: Mon, 22 Apr 2013 20:26:36 GMT
6 Server: gws
7 Content-Length: 218
8 X-XSS-Protection: 1; mode=block
9 X-Frame-Options: SAMEORIGIN
```

Listing 2: Html Header

And this is the expected html response definition from a successful request.

```

1 Response = Status-Line
2           *(( general-header
3              | response-header
4              | entity-header ) CRLF)
5           CRLF
6           [ message-body ]
```

Listing 3: Http Response

2.4.3 GET

A **GET** method is defined in the specification, to be used for operations that require **reading** information. On the server side, this should not tamper with the resources. A get request requires **Request-Line** field to be sent. It is up to the server to look for the resource requested, and send the response afterwards.

Taken directly from the manual, here is a demonstration of what the GET request looks like, when “accessing `www.w3.org`”:

```
1 GET /pub/WWW/TheProject.html HTTP/1.1
2 Host: www.w3.org
```

Listing 4: Http Get Request

2.4.4 POST

POST is the non-indempotent method we mentioned before. This means each time a post request is sent, a resource must be created, always guaranteeing different response, for a same request. An example for this would be creating a book with common data fields such as *author*, *year*, *title*, *etc...*

This is what a typical POST request looks like:

In this case study, we will be using this to write a text file on the server, with the request body as file contents. It would be possible to retrieve these contents later on.

2.4.5 PUT

PUT is a indempotent method. It is typically used in order to update resources on the sever side. For example, if we wish to update the book re

In this case study we will be using this to change a variable on the server. That way it will be possible to start the server, change this value, and get a response showing that the update was a success. It is quite possible to use this along with AUnit in order unit test these features automatically.

2.4.6 DELETE

DELETE is an indempotent method. It is typically used in order to delete a resource on a website. We will be using this to delete resources that are created by the POST definition of Axios (refer to 2.4.4).

2.5 First Steps

The first steps in an Ada application is to define the *project manager file* for the given project as outlined on the manual [2].

The GNAT project file abstracts a lot of the building details into this file. It will take care even about other language builds, meaning that if you wish to create, for example C extensions to the language, you could achieve this easily, as opposed to tedious makefiles.

Another added advantage is that GNAT Project Files also take into consideration the file naming conventions. Another automated feature in this file is automatic building of external libraries. There are various other ways to achieve this, either on a makefile level, or alternatively using features in certain SCM tools to clone and build the libraries automatically.

A notable feature of the GPR files, is that hierarchical builds are allowed, meaning that if different components require different builds, and different dependencies, that build can be isolated to that specific build alone [2]. The hierarchical layering of the building process, also exploits a similar aspect to that of a layered software architecture: different layers can be developed at parallel, as could subsystems.

2.5.1 Deployment View

The overall view of the system components can be seen in Figure 1.




Figure 1: Deployment Diagram

src	The directory to contain the sources
obj	The directory to contain the compiler output
bin	The directory to contain the binaries
obj/debug	The non-optimized compiled binaries with debug info are stored here. You can notice this in the GPR file, where the ‘-g’ flag is given to the compiler.
obj/release	The optimized compiled binaries without debug information. You can notice this in the GPR file, where the ‘O2’ flag is given to the compiler.
tests	The directory to contain the tests

Table 2: Directory Structure Specification for GPR File

Implementation

::Section 3::

This section will deal with the more technical aspects of this case study. From now on we will list topics that have to do with the implementation of the simple Http server.

3.1 Defining the required GNAT Project File Specification

The specifications list the requirements as noted on Table 2.

3.1.1 GNAT Project File

Here is the project file we will be using.

```

1 -- Simon Symeonidis, 2013
2 -- GNAT Project file for the Http server case study.
3 --
4 -- Used Manuals:
5 --   http://docs.adacore.com/gprbuild-docs/html/gprbuild-ug.html
6 --   http://docs.adacore.com/gnat-unw-docs/html/gnat-ugn-12.html
7 --
8 -- Example use:
9 --   gnatmake -P axios.gpr -Xmode=debug -p
10 project Axios is
11
12   -- Standard configurations
13   for Main use ("main.adb");
14   for Source_Dirs use ("src/**");
15   for Exec_Dir use "bin/";
16
17   -- Ignore git scm stuff
18   for Ignore_Source_Sub_Dirs use (".git/");
19
20   -- Objects are contained in their own directories (this is also
21   -- known as an isolated build).
22   for Object_Dir use "obj/" & external ("mode", "debug");
23   for Object_Dir use "obj/" & external ("mode", "release");
24
25   package Builder is
26     for Executable ("main.adb") use "axios";
27   end Builder;
```

```

28
29  -- To invoke either case, you need to set the -X flag at gnatmake in command
30  -- line. You will also notice the Mode_Type type. This constrains the values
31  -- of possible valid flags.
32  type Mode_Type is ("debug", "release");
33  Mode : Mode_Type := external ("mode", "debug");
34  package Compiler is
35    -- Either debug or release mode
36    case Mode is
37      when "debug" =>
38        for Switches ("Ada") use ("-g");
39      when "release" =>
40        for Switches ("Ada") use ("-O2");
41    end case;
42  end Compiler;
43
44  -- Not needed for this study, but listed
45  package Binder is
46  end Binder;
47
48  -- Not needed for this study, but listed
49  package Linker is
50  end Linker;
51
52 end Axios;

```

Listing 5: GNAT Project File

Notice that some of the flags for the compiler in both release, and debug modes are the same flags used for GNU C++ compiler. We explain each section with a little more detail.

Directory Structure The directory structure is as outlined on the previously demonstrated table. We see these defined in the GPR file by the names of **Source_Dirs**, **Exec_Dir**. Because of the definition of the Object file directories for a debug and release version, we also get the “obj/release” and “obj/debug” directories, as previously required. A Simple hello world application would yield the following structure, for example:

```

1  .
2  bin
3  bin/hello_world
4  hello.gpr
5  obj
6  obj/debug
7  obj/debug/main.ali
8  obj/debug/main.o
9  obj/release
10 obj/release/main.ali
11 obj/release/main.o
12 src
13 src/main.adb
14
15 5 directories, 7 files

```

Listing 6: Directory Structure

Flags There are many flags that can be set for the compilation and linking process. The manual covers a lot of them, but the ones used are explained.

Builder Notice the "package Builder is ..." line. We define the flags that should be used by the builder (gnatmake), when invoked.

Executable This is to specify the output name of the binary once the whole project is compiled.

Compiler Notice the "package Compiler is ..." line. This contains the flags for the actual compiler. We have two flags '-g', and '-O2' that the seasoned GNU C++ compiler user would recognize instantly. The '-g' flag would be used to produce non-optimized binaries in order to contain extra information when debugging.

The '-O2' is for optimization, and that is the reason why it is in the release case. We define a Mode_Type which can either be one of the two literals 'debug', and 'release' in order to achieve this.

3.2 Components

We now describe the components of the project. Figure 2




Figure 2: Overall organization of software components of Axios

3.2.1 Main Entry Point

The main entry point is where this particular software begins execution. In the case of this application the first steps that are fired up on execution, is spawning two different listeners. Listener L_1 accepts connections on port 3000, and another Listener L_2 accepts connections on port 8000. We exploit this fact, and make them point to different directories as well; L_1 to $Dir = www1/$ and L_2 to $Dir = www2/$.

```

1 -- @author Simon Symeonidis
2 -- @date Fri May 3
3 with Ada.Text_IO; use Ada.Text_IO;
4 with Ada.Strings.Unbounded;
5
6 -- user
7 with Listeners, File_Utils;
8 procedure Main is
9
10  -- Pointers so we pass to the task types.
11  type Listener_Access is access Listeners.Listener;
12
13  task type Launch_Listener(L : Listener_Access) is
14    entry Start;
15    entry Stop;

```

```

16  end Launch_Listener;
17
18  task body Launch_Listener is
19  begin
20      accept Start;
21      L.Print_Info;
22      L.Listen;
23      accept Stop;
24  end Launch_Listener;
25
26  -- listeners. Notice the '
27  l1 : Listener_Access :=
28      new Listeners.Listener'(Port_Number => 3000,
29      Host_Name      => Ada.Strings.Unbounded.To_Unbounded_String("localhost"),
30      Shutdown      => false,
31      WS_Root_Path => Ada.Strings.Unbounded.To_Unbounded_String("./www1/"));
32
33  l2 : Listener_Access :=
34      new Listeners.Listener'(Port_Number => 8000,
35      Host_Name      => Ada.Strings.Unbounded.To_Unbounded_String("localhost"),
36      Shutdown      => false,
37      WS_Root_Path => Ada.Strings.Unbounded.To_Unbounded_String("./www2/"));
38
39  -- listener tasks
40  lt1 : Launch_Listener(L => l1);
41  lt2 : Launch_Listener(L => l2);
42
43  begin
44      -- Start listening for connections.
45      lt1.Start;
46      lt2.Start;
47
48      -- On Graceful end
49      lt1.Stop;
50      lt2.Stop;
51  end Main;

```

Listing 7: Main Entry Point

The listeners are defined in the data block, and later in the body, their process starts by calling their ‘Start’ procedure. Next we investigate what actually happens within the listeners.

3.2.2 Listener

Specification of the listeners.

```

1  -- @author Simon Symeonidis
2  -- @date
3  --
4  -- This is the listener for http connections to port 80.
5  -- This should delegate the request to some handler, and
6  -- execute the proper logic.
7  with
8      Ada.Text_IO
9      , Ada.Strings.Unbounded
10     , Ada.Strings.Unbounded.Text_IO

```

```

11 , Ada.Exceptions
12 , Ada.Streams
13 , Ada.Calendar
14 , Ada.Real_Time
15 , GNAT.Sockets
16 -- user
17 , Transaction_Handlers
18 ;
19
20 use
21   Ada.Exceptions
22 , GNAT.Sockets;
23
24 use type
25   Ada.Streams.Stream_Element_Count
26 , Ada.Streams.Stream_Element_Array
27 , Ada.Real_Time.Time
28 , Ada.Real_Time.Time_Span
29 ;
30
31 package Listeners is
32
33   -- This is basically the specification for web servers. The way this is done
34   -- allows us to create multiple listeners on the same machine.
35   type Listener is tagged record
36     Port_Number   : Integer range 0 .. 16#ffff#;
37     Host_Name     : Ada.Strings.Unbounded.Unbounded_String;
38     Shutdown      : Boolean := False;
39     WS_Root_Path  : Ada.Strings.Unbounded.Unbounded_String;
40   end record;
41
42   procedure Print_Info(This : Listener);
43   procedure Listen(This : Listener);
44   function Tiny_Name(This : Listener) return String;
45   function Response_Date return String;
46 end Listeners;

```

Listing 8: Simple Listener Specification

Implementation of the listeners.

```

1 -- @author Simon Symeonidis
2 -- @date   Mon May 6 2013
3 -- Implementation for the listener
4 --
5 -- Some nice references / supplementary material
6 --
7 -- + Simple http server in ruby using tcpserver
8 --   http://stackoverflow.com/questions/7540064/
9 -- + Using Ada get_line for unknown input size
10 --   http://www.radford.edu/~nokie/classes/320/stringio.html
11
12 package body Listeners is
13
14   -- Print to the command line information such

```

```

15  — as host and port number
16  procedure Print_Info(This : Listener) is
17  begin
18      Ada.Text_IO.Put_Line("Listener Info: ");
19      Ada.Text_IO.Put_Line("Port Number : "
20          & Integer'Image(This.Port_Number));
21      Ada.Text_IO.Put_Line("Hostname   : "
22          & Ada.Strings.Unbounded.To_String(This.Host_Name));
23      Ada.Text_IO.Put_Line("Root Dir.   : "
24          & Ada.Strings.Unbounded.To_String(This.WS_Root_Path));
25      Ada.Text_IO.New_Line;
26  end Print_Info;
27
28  function Tiny_Name(This : Listener)
29  return String is
30      Name : String :=
31          Ada.Strings.Unbounded.To_String(This.Host_Name) & "@" &
32          Integer'Image(This.Port_Number);
33  begin
34      return Name;
35  end Tiny_Name;
36
37  — Return the date as string, in the format specified by the RFC2616 manual.
38  — For now we just add a placeholder date.
39  — TODO need to fix this
40  function Response_Date
41  return String is
42  begin
43      return "Date: Tue, 20 Jan 2012 10:48:45 GMT";
44  end Response_Date;
45
46  — Listen forever. Graceful shutdown if it receives some signal.
47  procedure Listen(This : Listener) is
48      Socket    : Socket_Type;
49      Server     : Socket_Type;
50      Address    : Sock_Addr_Type;
51      Channel    : Stream_Access;
52      The_Host   : String := Ada.Strings.Unbounded.To_String(This.Host_Name);
53  begin
54
55      Address.Addr := Addresses (Get_Host_By_Name (The_Host), 1);
56      Address.Port := Port_Type(This.Port_Number);
57      Create_Socket(Server);
58
59      Set_Socket_Option(Server, Socket_Level, (Reuse_Address, True));
60      Bind_Socket(Server, Address);
61      Listen_Socket(Server);
62      — pass to other socket.
63      Ada.Text_IO.Put_Line
64          ("Successfully listening on: " &
65          Port_Type'Image(Address.Port));
66
67      Listening_Loop :
68      while not This.Shutdown loop
69          Accept_Socket(Server, Socket, Address);
70          Channel := Stream(Socket);
71          declare
72              CRLF      : String      := ASCII.CR & ASCII.LF;
73              LF         : Character := ASCII.LF;

```

```

74     Counter      : Integer      := 0;
75
76     Request      : Ada.Strings.Unbounded.Unbounded_String;
77     Chara        : Character;
78
79     RTime_Start   : Ada.Real_Time.Time := Ada.Real_Time.Clock;
80     RTime_Stop    : Ada.Real_Time.Time;
81     RTime_Total   : Ada.Real_Time.Time_Span;
82 begin
83     — Read the request body
84     Read_Request :
85     loop
86     Character'Read(Channel, Chara);
87     Ada.Strings.Unbounded.Append(
88         Source => Request,
89         New_Item => Chara);
90
91     if Chara = ASCII.LF or Chara = ASCII.CR then
92         Counter := Counter + 1;
93     else — reset
94         Counter := 0;
95     end if;
96
97     exit when Counter = 4;
98 end loop Read_Request;
99
100    RTime_Stop := Ada.Real_Time.Clock;
101    RTime_Total := RTime_Stop - RTime_Start;
102
103    Ada.Text_IO.Put_Line(
104        Ada.Strings.Unbounded.To_String(
105            Request));
106
107    String'Write(Channel,
108        Transaction_Handlers.Handle_Request(
109            Ada.Strings.Unbounded.To_String(Request),
110            Ada.Strings.Unbounded.To_String(This.WS_Root_Path)));
111
112    exception when E : others =>
113        Ada.Text_IO.Put_Line
114            ("Listeners, at main listen loop: " &
115            Exception_Name(E) & Exception_Message(E));
116
117    end;
118    Free(Channel);
119    Close_Socket(Socket);
120 end loop Listening_Loop;
121
122 end Listen;
123 end Listeners;

```

Listing 9: Simple Listener Body

Next we investigate the transaction handlers. The transaction handlers help us generate the content that we need to relay to the clients (browsers).

3.2.3 Transaction Handlers

Specification of the transaction handlers.

```

1 with
2   Request_Helpers
3 , Response_Helpers
4 , File_Utils
5 ;
6
7 -- @author Simon Symeonidis
8 -- @date   Fri May 17
9 -- This encapsulates business logic of the http server (what to do
10 -- with each given request).
11 package Transaction_Handlers is
12   function Handle_Request(R : String; Context : String) return String;
13 end Transaction_Handlers;
```

Listing 10: Transaction Handler Specification

It's pretty straightforward - when the client (browser) sends us a request, we want to identify what kind of request that is, and what to do with it. Therefore we streamline this to the function 'Handle_Request'. Now let's take a look how we actually handle the implementation of the request.

Implementation of the transaction handlers.

```

1 -- @author Simon Symeonidis
2 -- Implementation of the transaction handlers
3 package body Transaction_Handlers is
4   function Handle_Request(R : String; Context : String)
5     return String is
6     package rh renames Response_Helpers;
7     package rr renames Request_Helpers;
8     R_Type : rr.Request_Type :=
9       rr.Parse_Request_Type(R);
10    URI      : String := rr.Parse_Request_URI(R);
11    begin
12    case R_Type is
13      when rr.GET =>
14        return
15          rh.Make_Response(
16            File_Utils.Read(Context & "/" & URI));
17
18      when rr.POST =>
19        return
20          rh.Make_Response("<h1>Got Post</h1>");
21
22      when rr.PUT =>
23        return
24          rh.Make_Response("<h1>Got Put</h1>");
25
26      when rr.DELETE =>
27        return
28          rh.Make_Response("<h1>Got DELETE</h1>");
29
```



```

30     when rr.HEAD =>
31         return
32         rh.Make_Response("<h1>Got HEAD</h1>");
33
34     when others =>
35         return
36         rh.Make_Response("<h1>Error!</h1>"
37         & "<p> You did something very weird in order to see this</p>");
38
39     end case;
40     end Handle_Request;
41 end Transaction_Handlers;

```

Listing 11: Transaction Handler Implementation

We see that we can handle the requests ‘GET’, ‘POST’, ‘PUT’, ‘DELETE’, ‘HEAD’ appropriately using the case clause. We’ve implemented functionality mainly for ‘GET’, since ‘Axios’ simply shares static information. We support however the rest of the ‘Http’ methods by returning simple strings to handle the requests at a minimum. Next we investigate the ‘Request_Helpers’ and ‘Reponse_Helpers’.

3.2.4 Request Helpers

Specification of request helpers.

```

1 with GNAT.Regpat;
2 with Ada.Text_IO;
3 -- @author Simon Symeonidis
4 -- @date Thu May 16 2013
5 package Request_Helpers is
6     type Request_Type is (HEAD, GET, POST, PUT, DELETE, OTHER);
7
8     function Parse_Request_Type(R : String) return Request_Type;
9     function Parse_Request_URI(R : String) return String;
10    procedure Get_Word(S
11                    : String;
12                    First, Last : out Positive;
13                    Found      : out Boolean);
14 end Request_Helpers;

```

Listing 12: Request Helper Specification

Implementation of request helpers.

```

1 -- @author Simon Symeonidis
2 -- @date Thu May 16 2013
3 package body Request_Helpers is
4     -- Parse the request type (get first line, see what type of
5     -- http method it is and return the valid enumeration).
6     function Parse_Request_Type(R : String)
7     return Request_Type is
8         First, Last : Positive;
9         Found      : Boolean;
10    begin

```

```

11   Get_Word(R, First, Last, Found);
12
13   if      R(First..Last) = "GET"      then
14       return GET;
15   elsif R(First..Last) = "POST"     then
16       return POST;
17   elsif R(First..Last) = "PUT"      then
18       return PUT;
19   elsif R(First..Last) = "HEAD"     then
20       return HEAD;
21   elsif R(First..Last) = "DELETE"  then
22       return DELETE;
23   else
24       return GET;
25   end if;
26 end Parse_Request_Type;
27
28 — This extracts the path requested from the first line in the http
29 — method eg:
30 —
31 —   GET /index.html
32 —   ...
33 —
34 — Would return a string with "index.html"
35 function Parse_Request_URI(R : String)
36 return String is
37     First, Last : Positive;
38     Found : Boolean;
39 begin
40     — Called twice, because the second 'word' is the uri in the request
41     — header.
42     Get_Word(R, First, Last, Found);
43     Get_Word(R(Last+1..R'Last), First, Last, Found);
44
45     if Found then
46         return R(First..Last);
47     end if;
48
49     return "uri not found.";
50 end Parse_Request_URI;
51
52 — @note Thanks to
53 —   http://rosettacode.org/wiki/Regular\_expressions#Ada
54 procedure Get_Word(S           : String;
55                   First, Last : out Positive;
56                   Found       : out Boolean) is
57     Pattern      : constant String := "(\\S+)";
58     Compiled     : GNAT.Regpat.Pattern_Matcher :=
59         GNAT.Regpat.Compile(Pattern);
60     Match_Array  : GNAT.Regpat.Match_Array(0..1);
61 begin
62     GNAT.Regpat.Match(Compiled, S, Match_Array);
63     Found := not GNAT.Regpat."="(Match_Array(1), GNAT.Regpat.No_Match);
64
65     if Found then
66         First := Match_Array(1).First;
67         Last  := Match_Array(1).Last;
68     end if;
69 end Get_Word;

```

```
70 end Request_Helpers;
```

Listing 13: Request Helper Implementation

Here we see that the *Request Helper* provides us with some auxiliary functions that we can use in order to discern what kind of request is incoming. We are also provided with another auxiliary function to extract the requested path from ‘GET’ Http methods. We use these paths in order to look into the static resources, and if found, send the resource back to the user. So for example L_1 has a directory *www1*, so the path requested would be prefixed with *www1*, and return the contents if found. Else an error message is served back.

3.2.5 Response Helpers

Specification of response helpers.

```
1 -- @author Simon Symeonidis
2 -- @date   Fri May 17
3 -- This is a helper for forming the responses.
4 package Response_Helpers is
5   function Headers return String;
6   function Response_Date return String;
7   function Make_Response(S : String) return String;
8 private
9   CRLF : constant String := ASCII.CR & ASCII.LF;
10 end Response_Helpers;
```

Listing 14: Response Helper Specification

The two main important functions that one should notice are ‘Headers’ and ‘Make_Response’. ‘Header’ is to ready a header for us to send back, and the ‘Make_Response’ is to combine a header, with the required HTML body, and calculate a length to provide to the ‘Content-Length’ http header field.

Implementation of response helpers.

```
1 with Ada.Calendar;
2 with GNAT.Calendar.Time_IO;
3 -- @author Simon Symeonidis
4 -- This is the response helper that readies a header to send back to
5 -- the client and concatenates the html body as well.
6 package body Response_Helpers is
7
8   -- @note don't use this directly
9   function Headers
10  return String is
11  begin
12    return
13      "Http/1.1 200 OK" & CRLF &
14      Response_Date & CRLF &
15      "Server: axios" & CRLF &
16      "Content-Type: text/html; charset=iso-8859-1" & CRLF &
17      "Content-Length: ";
18  end Headers;
```

```

19
20 function Response_Date
21 return String is
22     package ac renames Ada.Calendar;
23     package gct renames GNAT.Calendar.Time_IO;
24     Current_Time : ac.Time;
25     Format        : gct.Picture_String := "%a, %d %B %Y %H:%M:%S EST";
26     Field_Name    : constant String   := "Date: ";
27 begin
28     Current_Time := ac.Clock;
29     return
30         Field_Name &
31         gct.Image(Current_Time, Format);
32 end Response_Date;
33
34 -- Make the response for the browser. This function includes
35 -- the headers, (including the content-length field of the body),
36 -- and the message body.
37 function Make_Response(S : String)
38 return String is
39 begin
40     return
41         Headers &
42         Positive'Image(S'Length) & CRLF & CRLF &
43         S;
44 end Make_Response;
45
46
47 end Response_Helpers;

```

Listing 15: Response Helper Implementation

And here we see the realization of the specification. ‘Header’ indeed provides a function to construct an http header to send back. This method is used along with ‘Make.Respone’ in order to send back everything with the required body containing the ‘Html’. The body is in fact the requested resource from the user. Last but not least, we present some file utilities that help us achieve sending back the required resources.

3.2.6 File Utils

Specification of file utils.

```

1 with Ada.Strings.Unbounded;
2 -- @author Simon Symeonidis
3 -- Specification of the file utils helper
4 package File_Utils is
5     procedure
6         Write(File_Name : String; Contents : String);
7
8     procedure
9         Append(File_Name : String; Contents : String);
10
11     function
12         Read(File_Name : String)
13         return String;
14 end File_Utils;

```

Listing 16: File Utils Specification

Implementation of file utils.

```

1  -- @author Simon Symeonidis
2  --
3  -- Common read and write utils so that we separate the job of the
4  -- post and put method calls to this package. This is also a shorthand
5  -- for pretty basic file read and write operations.
6  --
7  -- Thanks to
8  -- http://rosettacode.org/wiki/Read\_entire\_file#Ada.Direct\_IO
9  -- http://rosettacode.org/wiki/Read\_a\_file\_line\_by\_line#Ada
10 -- http://faculty.cs.wvu.edu/reedyc/AdaResources/LectureNotes/Text\_Files.html
11 -- http://en.wikibooks.org/wiki/Ada\_Programming/Libraries/Ada.Text\_IO
12
13 with Ada.Text_IO;
14 with Ada.Direct_IO;
15 with Ada.Exceptions; use Ada.Exceptions;
16
17 package body File_Utils is
18
19   -- Write to a file with bounded strings. Please note this will overwrite
20   -- previously created files.
21   procedure Write(File_Name : String; Contents : String) is
22     The_File_Mode : Ada.Text_IO.File_Mode := Ada.Text_IO.Out_File;
23     The_File      : Ada.Text_IO.File_Type;
24   begin
25     Ada.Text_IO.Create
26       (File => The_File,
27        Mode => The_File_Mode,
28        Name => File_Name);
29     Ada.Text_IO.Put
30       (File => The_File,
31        Item => Contents);
32     Ada.Text_IO.Close(File => The_File);
33
34   exception when E : others =>
35     Ada.Text_IO.Put_Line
36       (Exception_Name(E) & Exception_Message(E));
37   end Write;
38
39   -- This will read the full contents of a given file.
40   -- Read a file by name, and then return the file text contents, lines
41   -- delimited by a ASCII.LF
42   function Read(File_Name : String)
43   return String is
44     The_File_Mode : Ada.Text_IO.File_Mode := Ada.Text_IO.In_File;
45     The_File      : Ada.Text_IO.File_Type;
46     Contents      : Ada.Strings.Unbounded.Unbounded_String;
47   begin
48     Ada.Text_IO.Open
49       (File => The_File,

```

```

50     Mode => The_File_Mode ,
51     Name => File_Name);
52
53     while not Ada.Text_IO.End_Of_File (The_File) loop
54         declare
55             Line : String := Ada.Text_IO.Get_Line(The_File);
56         begin
57             Ada.Strings.Unbounded.Append(
58                 Source => Contents, New_Item => Line(1..Line'Last));
59         end;
60     end loop;
61
62     Ada.Text_IO.Close(File => The_File);
63     return Ada.Strings.Unbounded.To_String(Contents) ;
64
65     exception when E : others =>
66         Ada.Text_IO.Put_Line("Warning: request to a non existant file was made.");
67         Ada.Text_IO.Put_Line(">> Path: " & File_Name);
68         return "Error";
69     end Read;
70
71     -- Append to file by given name.
72     procedure Append(File_Name : String; Contents : String) is
73     begin
74         null;
75         --exception when E : others =>
76         -- Ada.Text_IO.Put_Line
77         -- (Exception_Name(E) & Exception_Message(E));
78     end Append;
79
80 end File_Utils;

```

Listing 17: Request Helper Implementation

Mainly the ‘Read’ function is what is of interest here. But this package is straightforward, and does not need any more explanations.

3.3 Testing

Testing is achieved by using **AUnit**, that provides us with some facilities. More information can be read here [3].

Demonstration

Section 4:

Figure 3 is a small demo on what it looks like when accessing the *AXIOS* server. Figure 4 shows the output of the server (which is the text request sent by the browser, unless there is an error raised).




Figure 3: Using a browser to access the server

```

[psyonn@aeolus bin 0]$ ls
axios www1 www2
[psyonn@aeolus bin 0]$ ls
axios www1 www2
[psyonn@aeolus bin 0]$ ./axios
Listener Info:
Port Number : 8000
Hostname    : localhost
Root Dir.   : ./www2/

Successfully listening on: 8000
Listener Info:
Port Number : 3000
Hostname    : localhost
Root Dir.   : ./www1/

Successfully listening on: 3000
GET /index.html HTTP/1.1
Host: localhost:3000
Connection: keep-alive
Accept: text/html,application/xhtml+xml,application/xml;q=0.9,*/*;q=0.8
User-Agent: Mozilla/5.0 (X11; Linux x86_64) AppleWebKit/537.31 (KHTML, like Gecko) Chrome/26.0.1410.63 Safari/537.31
Accept-Encoding: gzip,deflate,sdch
Accept-Language: en-US,en;q=0.8,el;q=0.6
Accept-Charset: ISO-8859-1,utf-8;q=0.7,*;q=0.3

GET /favicon.ico HTTP/1.1
Host: localhost:3000
Connection: keep-alive
Accept: */*
User-Agent: Mozilla/5.0 (X11; Linux x86_64) AppleWebKit/537.31 (KHTML, like Gecko) Chrome/26.0.1410.63 Safari/537.31
Accept-Encoding: gzip,deflate,sdch
Accept-Language: en-US,en;q=0.8,el;q=0.6
Accept-Charset: ISO-8859-1,utf-8;q=0.7,*;q=0.3

```

Figure 4: Server output when receiving requests

Appendix

The appendix will give some extra simplified information which might help understand the smaller parts of the given case study. Explanations and code listings are provided as best as possible.

5.1 Networking

A lot of the networking will seem very familiar to you if you have used BSD C Sockets. Provided are two Ada procedures. One listener, and one sender. The Listener listens for incoming connections. If a message is intercepted, the message is uppercased, and sent back to the client. The client simply sends a string message to the listener. The listener listens at port 3000 (it is assumed these procedures run on your local machine, together).

```

1  -- @author Simon Symeonidis
2  -- A concise example of a listener procedure, adapted from the
3  -- ping pong example in the GNAT.Sockets specification file.
4  with Ada.Text_IO;
5  with Ada.Strings;
6  with Ada.Strings.Fixed;
7  with Ada.Strings.Maps;
8  with Ada.Strings.Maps.Constants;
9  with GNAT.Sockets; use GNAT.Sockets;
10 with Ada.Exceptions; use Ada.Exceptions;
11
12 procedure Simple_Listener is
13   Address   : Sock_Addr_Type;
14   Server    : Socket_Type;
15   Socket    : Socket_Type;
16   Channel   : Stream_Access;
17 begin
18   -- Init socket stuff.
19   Address.Addr := Addresses (Get_Host_By_Name (Host_Name), 1);
20   Address.Port := 3000;
21   Create_Socket (Server);
22
23   Set_Socket_Option(Server, Socket_Level, (Reuse_Address, True));
24   Bind_Socket (Server, Address);
25   Listen_Socket(Server);
26   Accept_Socket(Server, Socket, Address);
27   Channel := Stream (Socket);
28   declare
29     Message : String := String'Input (Channel);
30   begin
31     Ada.Text_IO.Put_Line("Got: " & Message);
32     String'Output (Channel,
33       Ada.Strings.Fixed.Translate(Message,
34         Ada.Strings.Maps.Constants.Upper_Map));
35   end;
36
37   Close_Socket (Server);
38   Close_Socket (Socket);
39
40   exception when E : others =>
41     Ada.Text_IO.Put_Line
42       (Exception_Name (E) & Exception_Message(E));

```

```
43 end Simple_Listener;
```

Listing 18: Simple Listener

```
1 -- @author Simon Symeonidis
2 -- A concise example of a client procedure, adapted from the
3 -- ping pong example in the GNAT.Sockets specification file.
4
5 with Ada.Text_IO;
6 with GNAT.Sockets; use GNAT.Sockets;
7
8 procedure Simple_Sender is
9   Address : Sock_Addr_Type;
10  Socket   : Socket_Type;
11  Channel  : Stream_Access;
12 begin
13   -- Init socket stuff
14   Address.Addr := Addresses (Get_Host_By_Name (Host_Name), 1);
15   Address.Port := 3000;
16   Create_Socket(Socket);
17   Set_Socket_Option(Socket, Socket_Level, (Reuse_Address, True));
18   Connect_Socket (Socket, Address);
19   Channel := Stream (Socket);
20   String'Output (Channel, "hack the planet.");
21
22   declare
23     Message : String := String'Input(Channel);
24   begin
25     Address := Get_Address(Channel);
26     Ada.Text_IO.Put_Line ("Server @ " & Image(Address)
27                           & ": " & Message);
28   end;
29
30   -- Cleanup
31   Close_Socket(Socket);
32 end Simple_Sender;
```

Listing 19: Simple Sender

Also, here is an example http get program that helped a lot in this case study. It is one of the GNAT Sockets examples that can be found in the following location [1]. I have modified the source a little, and provided it in listing 20.

```
1 with Ada.Text_IO;           use Ada.Text_IO;
2 with GNAT.Sockets;         use GNAT.Sockets;
3 with Ada.Strings.Unbounded; use Ada.Strings.Unbounded;
4
5 with Ada.Streams;
6 use type Ada.Streams.Stream_Element_Count;
7
8 -- Thanks to
```

```

9  — http://en.wikibooks.org/wiki/Ada\_Programming/Libraries/Ada.Streams#
   Read_attribute
10 procedure Vcv is
11   Client  : Socket_Type;
12   Address : Sock_Addr_Type;
13   Channel : Stream_Access;
14
15   CRLF : String := (1 => ASCII.CR, 2 => ASCII.LF);
16   Send : String := CRLF & CRLF;
17   Offset : Ada.Streams.Stream_Element_Count;
18   Data : Ada.Streams.Stream_Element_Array (1 .. 256);
19 begin
20   GNAT.Sockets.Initialize; — initialize a specific package
21   Create_Socket (Client);
22   Address.Addr := Addresses (Get_Host_By_Name("www.google.com"));
23   Address.Port := 80;
24
25   Connect_Socket (Client, Address);
26   Channel := Stream (Client);
27
28   String'Write (Channel, "GET / HTTP/1.1" & CRLF &
29                     "Connection: close " & Send);
30   loop
31     Ada.Streams.Read (Channel.All, Data, Offset);
32     exit when Offset = 0;
33     for I in 1 .. Offset loop
34       Ada.Text_IO.Put (Character'Val (Data (I)));
35     end loop;
36   end loop;
37 end Vcv;

```

Listing 20: VCV Get Http Client

5.2 Task Oriented Concurrency Model

Concurrency in Ada uses tasks. Here is a small example to demonstrate this feature of the language that exploits a very interesting way of achieving concurrent tasks. If more detail is required, refer to [10]

5.2.1 Simple Concurrent Program

This is the simplest example in using tasks in Ada. The example was adapted from notes found in [6], and was also the first concurrent program I wrote for testing and understanding tasks, the first time around.

```

1  with Ada.Text_IO;
2
3  procedure Task_Types is
4   task type A (Label : Character) is
5     entry Start;
6   end A;
7
8   task body A is
9   begin
10    accept Start;
11
12    for I in Integer range 1..10 loop

```

```

13     Ada.Text_IO.Put_Line(Character'Image(Label) & " is working...");
14     delay 0.4;
15 end loop;
16 end A;
17
18 My_Task_01 : A(Label => 'A');
19 My_Task_02 : A(Label => 'B');
20 My_Task_03 : A(Label => 'C');
21
22 begin
23   My_Task_01.Start;
24   My_Task_02.Start;
25   My_Task_03.Start;
26 end Task_Types;

```

Listing 21: Task Types in Ada

This procedure would have the following output:

```

1 [psyomn@aeolus concurrency 0]$ ./task_types
2 'A' is working...
3 'B' is working...
4 'C' is working...
5 'A' is working...
6 'B' is working...
7 'C' is working...
8 'A' is working...
9 'B' is working...
10 'C' is working...
11 'A' is working...
12 'C' is working...
13 'B' is working...
14 'A' is working...
15 'B' is working...
16 'C' is working...
17 'B' is working...
18 'C' is working...
19 'A' is working...
20 'A' is working...
21 'B' is working...
22 'C' is working...
23 'A' is working...
24 'B' is working...
25 'C' is working...
26 'A' is working...
27 'B' is working...
28 'C' is working...
29 'A' is working...
30 'B' is working...
31 'C' is working...

```

Listing 22: Output for the Task Types in Ada

5.3 Text Manipulation

We list some text manipulation example code in this section, if better understanding is required. First it would be better to familiarize oneself with the *String* type; one can refer to [9], and another excellent tutorial of Java vs Ada String type comparison [7].

5.3.1 Splitting

A common task one wants to do with strings is split them. Refer to this [5], to read upon string splitting. Notice that this was added on the GNAT platform.

5.3.2 Regular Expressions

Regular expressions don't quite exist in the Ada programming language. However the GNAT platform provides some libraries with similar effects. A simple example can be found at [4].

5.4 Random Notes

Some other miscellaneous random notes, that might help in the future.

5.4.1 ADS / ADB file location

If the *gcc-ada* compiler is being used, the specification and body files can typically be found in the following directory

```
/usr/lib/gcc/$MACHTYPE/$GCC - VERSION/adainclude/
```

On another platform, namely *Lubuntu*, the absolute path included another directory right before the ada include one.

```
/usr/lib/gcc/$MACHTYPE/$GCC - VERSION/rti-native/adainclude/
```

5.4.2 Sanity Checking

I wanted to test out if things were working properly every now and then. Here are a few commands that I used in order to do this.

5.4.3 Using Telnet to connect to a Http Server

You can use telnet to connect to a http server, and write the headers manually, and receive the actual response. This helps at understanding a little more how the protocol works, and was used in order to debug the server.

```
1 [psyomn@aeolus ~ 0]$ telnet www.concordia.ca 80
2 Trying 132.205.244.34...
3 Connected to www.concordia.ca.
4 Escape character is '^'.
5 HEAD / Http/1.1
6 Host: www.concordia.ca
7 Connection: close
8
9 HTTP/1.1 200 OK
10 Date: Sun, 28 Apr 2013 23:08:49 GMT
11 Server: Apache
12 X-Powered-By: PHP/5.1.6
13 Vary: Accept-Encoding
14 Connection: close
```

```

15 Content-Type: text/html; charset=UTF-8
16
17 Connection closed by foreign host.

```

Listing 23: Telnet to an Http Server

5.4.4 Using ncat to dump packets to a binary file

On the various stages of debugging, *ncat* was used in order to dump the packets to a file, in order for inspection. This helped when I was facing trouble when the socket would send random garbage due to mistakes to the http servers, and have undesired consequences.

```

1 # First we start ncat on localhost, and make it listen at port 3000
2 [psyomn@aeolus ~ 0]$ ncat -l localhost 3000 > log.txt
3
4 # Then we telnet to ncat, at port 3000
5 [psyomn@aeolus ~ 0]$ telnet localhost 3000
6 Trying 127.0.0.1...
7 Connected to localhost.
8 Escape character is '^]'.
9 Hello there.
10 These are messages that are sent to you via telnet
11 I hope you like them.
12 Connection closed by foreign host.
13
14 # The log has the messages sent. The log file can also store byte values, not
15 # just text messages, hence a means to see what the application is writing at
16 # said sockets.
17 [psyomn@aeolus ~ 0]$ cat log.txt
18 Hello there.
19 These are messages that are sent to you via telnet
20 I hope you like them.

```

Listing 24: Using ncat to dump packets

Here is an investigation example of data sent through sockets from a test script. This is a GET request.

```

1 [psyomn@aeolus 423 0]$ xxd /tmp/loggy
2 00000000: 4745 5420 2f20 4874 7470 2f31 2e31 0d0a  GET / Http/1.1..
3 00000010: 486f 7374 3a20 6c6f 6361 6c68 6f73 740d  Host: localhost.
4 00000020: 0a43 6f6e 6e65 6374 696f 6e3a 2063 6c6f  .Connection: clo
5 00000030: 7365 0d0a 0d0a                                     se....

```

Listing 25: Using ncat a test script and xxd to hex dump

You can notice the CRLFs at the end of each request (that is the two bytes on the left, 0x0A, and 0x0D respectively).

And here is a small, priceless one-liner that has helped me debug some of my code during this case study. It uses *ncat*, and pipes the output to *xxd*, which in turn makes a hex dump.

```

1 [psyomn@aeolus 423 0]$ for (( ;; )) do ncat -l localhost 8080 | xxd; done
2 00000000: 0100 0000 0500 0000 4865 6c6c 6f          .....Hello
3 00000000: 0100 0000 0100 0000 48                      .....H
4 00000000: 0100 0000 0100 0000 48                      .....H
5 00000000: 4745 5420 2f20 4874 7470 2f31 2e31 0d0a  GET / Http/1.1..
6 00000010: 486f 7374 3a20 6c6f 6361 6c68 6f73 740d  Host: localhost.
7 00000020: 0a43 6f6e 6e65 6374 696f 6e3a 2063 6c6f  .Connection: clo
8 00000030: 7365 0d0a 0d0a                                se....
9 00000000: 2a00 0000 0100 0000 0100 0000 42          *.....B
10 00000000: 2a00 0000 0100 0000 0100 0000 42          *.....B
11 00000000: 2a00 0000                                *...
12 00000000: 2a00 0000 2a00 0000 2a00 0000          *...*...*...

```

Listing 26: One liner to hex dump dynamically

References

- [1] GNAT Sockets examples, 2013. [Online Fri May 3 2013].
- [2] adacore. 2013. http://docs.adacore.com/gnat-unw-docs/html/gnat_ugn_12.html.
- [3] adacore. AUnit, Ada Testing, 2013. http://docs.adacore.com/aunit-docs/aunit.html#AUnit_002eSimple_005fTest_005fCases.
- [4] Rosetta Code. Regular Expressions for Ada, 2013. http://rosettacode.org/wiki/Regular_expressions#Ada.
- [5] commons.ada.cx. Substrings using Gnat.String.Split, 2013. <http://bit.ly/107SVuN>.
- [6] M. J. Feldman. Chapter 15: Introduction to Concurrent Programming, 1996. <http://www.seas.gwu.edu/~mfeldman/cs2book/chap15.html>.
- [7] Edward G. (Ned) Okie. Differences Between Strings in Ada and Java, 2013. <http://www.radford.edu/~nokie/classes/320/strings.html>.
- [8] The Internet Society. 1999. <http://tools.ietf.org/pdf/rfc2616.pdf>.
- [9] Wikibooks. Ada Programming: String Type, 2013. http://en.wikibooks.org/wiki/Ada_Programming/Strings.
- [10] Wikibooks. Ada Programming: Tasking, 2013. http://en.wikibooks.org/wiki/Ada_Programming/Tasking.